4

KURBEN ADIERAZPENA FORMA PARAMETRIKOAN

4.1. Kurben parametrizazioa planoan

Forma parametrikoan emandako kurba baten ardatz koordenatu errektangeluar bidimentsionaleko OXY sistemako adierazpen grafikoa, OXY plano kartesiarrean (x(t),y(t)) bikoteak adieraztea da, t parametroa x(t) eta y(t) funtzioen izate eremuetan egonik.



▼ ParametricPlot[ ]

```
? ParametricPlot
```

ParametricPlot[{f_x, f_y}, {u, u_{min}, u_{max}}] generates a parametric plot of a curve with x and y coordinates f_x and f_y as a function of u.
ParametricPlot[{{f_x, f_y}, {g_x, g_y}, …}, {u, u_{min}, u_{max}}] plots several parametric curves.
ParametricPlot[{f_x, f_y}, {u, u_{min}, u_{max}}, {v, v_{min}, v_{max}}] plots a parametric region.
ParametricPlot[{{f_x, f_y}, {g_x, g_y}, …}, {u, u_{min}, u_{max}}, {v, v_{min}, v_{max}}] plots several parametric regions. ≫

```
Clear["Global`*"]
r[t_] = {x[t_], y[t_]} = {Cos[3 * t], Sin[2 * t]};
```

```
GraphicsArray[
 {Plot[x[t], {t, 0, 2 * Pi}, PlotStyle → {RGBColor[0.7, 0.3, 0.2], Thickness[0.01]}],
  Plot[y[t], {t, 0, 2 * Pi}, PlotStyle → {RGBColor[0.2, 0.3, 0.8], Thickness[0.01]}]}]
```




```
ParametricPlot[{Cos[3 * t], Sin[2 * t]}, {t, 0, 2 * Pi},
 PlotStyle → {RGBColor[0.2, 0.9, 0.2], Thickness[0.008]}, PlotRange → {{-1, 1}, {-1, 1}}]
```



▼ Manipulate[ ]

Agindu honek posible egiten du, t parametroak balio ezberdinak hartu ahala grafikoa nola irudikatzen den ikustea.

```
Manipulate[ParametricPlot[{Cos[3 * t], Sin[2 * t]},
  {t, 0, a}, PlotStyle → {RGBColor[0.2, 0.9, 0.2], Thickness[0.008]},
  PlotRange → {{-1, 1}, {-1, 1}}], {a, 0.2, 2 * Pi}]
```



4.2. Forma esplizituan emandako kurben parametrizazioa

Forma esplizituan definitutako y = f (x) funtzioa emanda, beti da posiblea hau era honetan parametrizatzea

x (t) =x(t)

y (t) = f (x(t))

★ Adibidea 1

```
{ParametricPlot[{u, Tan[u]}, {u, -2 Pi, 2 Pi},
  ExclusionsStyle → Dashed, Exclusions → {Cos[u] == 0}, PlotLabel → y == Tan[x]],
 ParametricPlot[{Tan[u], u}, {u, -2 Pi, 2 Pi}, ExclusionsStyle → Dashed,
  Exclusions → {Cos[u] == 0}, PlotLabel → y == ArcTan[x]]}
```



4.3. Forma inplizituan emandako funtzioen parametrizazioa

Forma inplizituan definitutako funtzioa emanda, beti da posiblea hau era honetan parametrizatzea

x (t) =x(t)

y (t)=y , hau f (x(t),y)=0 ekuazioaren soluzioa izanik

▼ (a,b) zentrudun eta r erradiodun zirkunferentzia baten parametrizazioa

```
zir = (x - a)^2 + (y - b)^2 == r^2
```

$(-a+x)^2 + (-b+y)^2 == r^2$

```
x[t_] = a + r * Cos[t]
```

$a + r\,\mathrm{Cos}[t]$

```
Solve[zir, y] /. x → x[t] // Simplify
```

$\left\{\left\{y \to b - \sqrt{r^2\,\mathrm{Sin}[t]^2}\right\}, \left\{y \to b + \sqrt{r^2\,\mathrm{Sin}[t]^2}\right\}\right\}$

```
zirkulu[t_, a_, b_, r_] = {x[t_], y[t_]} = {a + r * Sin[t], b + r * Cos[t]}
```

$\{a + r\,\mathrm{Sin}[t],\ b + r\,\mathrm{Cos}[t]\}$

▼ Zentrua jatorrian duen eta r erradiodun zirkunferentzia baten parametrizazioa

★ Orientazio positibodun lehenengo parametrizazioa

```
r[t_] = {x[t_], y[t_]} = {Cos[t], Sin[t]};
```

```
GraphicsArray[
 {Plot[Cos[t], {t, 0, 2 * Pi}, PlotStyle → {RGBColor[0.7, 0.3, 0.2], Thickness[0.01]}],
  Plot[Sin[t], {t, 0, 2 * Pi}, PlotStyle → {RGBColor[0.2, 0.3, 0.8], Thickness[0.01]}]}]
```



```
ParametricPlot[Evaluate[Table[{i Cos[u], i Sin[u]}, {i, 1, 3}]], {u, 0, 2 Pi}]
```



```
ParametricPlot[{i Cos[u], i Sin[u]}, {i, 1, 3}, {u, 0, 2 Pi}]
```



```
Manipulate[Show[ParametricPlot[{{Cos[t], Sin[t]}, {0.3 * Cos[t], 0.3 * Sin[t]}},
    {t, 0, a}, PlotStyle → {{RGBColor[0.2, 0.9, 0.5], Thickness[0.008]},
      {RGBColor[0.9, 0.2, 0.9], Thickness[0.008]}}, PlotRange → {{-1, 1}, {-1, 1}}],
  Graphics[Arrow[{{0, 0}, {Cos[a], Sin[a]}}]]], {a, 0.2, 2 * Pi}]
```



★ **Bigarren parametrizazioa: erloju-orratzen noranzkoan**

```
r[t_] = {x[t_], y[t_]} = {Cos[t], Sin[2 t]};

GraphicsArray[
 {Plot[Sin[t], {t, 0, 2 * Pi}, PlotStyle → {RGBColor[0.7, 0.3, 0.2], Thickness[0.01]}],
  Plot[Cos[t], {t, 0, 2 * Pi}, PlotStyle → {RGBColor[0.2, 0.3, 0.8], Thickness[0.01]}]}]
```



```
Manipulate[Show[ParametricPlot[{{Sin[t], Cos[t]}, {0.3 * Sin[t], 0.3 * Cos[t]}},
    {t, 0, a}, PlotStyle → {{RGBColor[0.2, 0.9, 0.5], Thickness[0.008]},
      {RGBColor[0.9, 0.2, 0.9], Thickness[0.008]}}, PlotRange → {{-1, 1}, {-1, 1}}],
  Graphics[Arrow[{{0, 0}, {Sin[a], Cos[a]}}]]], {a, 0.2, 2 * Pi}]
```



Astroidea

```
Clear[a]

astroidea[t_, a_] = {a * Cos[t]^3, a * Sin[t]^3};

ParametricPlot[{Cos[t]^3, Sin[t]^3}, {t, 0, 2 π}, AspectRatio → Automatic,
 PlotStyle → Flatten[Table[RGBColor[a, 0, c], {a, 0, 1}, {c, 0, 1}]],
 PlotLabel → Style["ASTROIDEA", Bold, Blue, 14], Frame → True]
```



```
Manipulate[ParametricPlot[{Cos[t]^3, Sin[t]^3}, {t, 0, a},
  AspectRatio → Automatic, PlotRange → {{-1, 1}, {-1, 1}}], {a, 0.01, 2 * Pi, 0.1}]
```



▼ Zikloidea

```
zikloidea[t_, a_] = a * {t - Sin[t], 1 - Cos[t]}
```

{a (t - Sin[t]), a (1 - Cos[t])}

```
c1 = ParametricPlot[{zikloidea[t, 1]}, {t, 0, 4 * Pi}, PlotStyle → RGBColor[0.6, 0, 0.2]]
```




```
c2 = ParametricPlot[{zikloidea[t, 2]}, {t, 0, 2 * Pi}, PlotStyle → RGBColor[0.2, 0.5, 0]]
```




```
Show[{c1, c2}, PlotRange → {0, 4}]
```



▼ Elipsea

```
Clear["Global`*"]
elipsea[t_, a_, b_, c_, d_] = {a * Sin[t], b * Cos[t]} + {c, d};
```

```
ParametricPlot[Evaluate[{elipsea[t, 4, 2, 0, 0], elipsea[t, 2, 3, 0, 0] + {1, 1}},
  {t, 0, 2 Pi}, AspectRatio → Automatic]]
```



4.4. Forma polarrean emandako kurben parametrizazioa

Forma polarrean definitutako r=r(t) funtzioa emanda, beti da posiblea hau era honetan parametrizatzea

x (t) =r(t) cost

y (t) =r(t) sent

Kardioidea

```
Clear["Global`*"]

kardioidea[t_, a_] = {a * Cos[t] * (1 + Cos[t]), a * Sin[t] (1 + Cos[t])};

ParametricPlot[kardioidea[t, 2], {t, 0, 2 Pi}]
```



Lemniskata

```
lemniskata[t_, a_] = {a * Cos[t] / (1 + Sin[t]^2), a * Sin[t] * Cos[t] / (1 + Sin[t]^2)};

Manipulate[
 ParametricPlot[{4 * Cos[t] / (1 + Sin[t]^2), 4 * Sin[t] * Cos[t] / (1 + Sin[t]^2)}, {t, 0, a},
  AspectRatio → Automatic, PlotRange → {{-5, 5}, {-1.5, 1.5}}], {a, 0.01, 2 * Pi, 0.1}]
```



```
lemniskata[t_, a_] = {a * Cos[t] / (1 + Sin[t]^2), a * Sin[t] * Cos[t] / (1 + Sin[t]^2)};
```

```
ParametricPlot[Evaluate[Table[lemniskata[t, a], {a, 1, 5}]], {t, 0, 2 π},
 AspectRatio → Automatic, PlotLabel → Style["LEMNISKATA FAMILIA", Bold, 14]]
```



▼ Espiral Logaritmikoa

```
espiralog[t_, a_, b_] = {a * E^(b * t) * Cos[t], a * E^(b * t) * Sin[t]}
```

$\{a\, e^{b\, t} \operatorname{Cos}[t],\ a\, e^{b\, t} \operatorname{Sin}[t]\}$

```
ParametricPlot[espiralog[t, 3, 0.05], {t, 0, 12 Pi}, AspectRatio → Automatic]
```



▼ Arkimedes-en espirala

```
ParametricPlot[{u Cos[u], u Sin[u]}, {u, 0, 4 Pi}, PlotStyle → Thick,
 ColorFunction → Function[{x, y, u, v}, Hue[u / (2 Pi)]], ColorFunctionScaling → False]
```



```
ParametricPlot[{(v + u) Cos[u], (v + u) Sin[u]}, {u, 0, 4 Pi},
 {v, 0, 5}, Mesh → {20, 5}, MeshShading → {{Purple, Cyan}, {Pink, Orange}}]
```



4.5. Adar infinitudun kurbak

▼ Deskartes-en folium-a

★ Parametrizazioaren definizioa

$$r[t\_] = \{x[t\_], y[t\_]\} = \left\{\frac{t}{1+t^3}, \frac{t^2}{1+t^3}\right\};$$

```
GraphicsArray[
 {Plot[x[t], {t, -3, 3}, PlotStyle → {RGBColor[0.7, 0.3, 0.2], Thickness[0.01]}],
  Plot[y[t], {t, -3, 3}, PlotStyle → {RGBColor[0.2, 0.3, 0.8], Thickness[0.01]}]}]
```





★ Ebaki puntuen eta adar infinituen azterketa

<u>Ebaki puntuak</u>

```
Solve[x[t] == 0, t]
```

```
{{t → 0}}
```

```
Solve[y[t] == 0, t]
```

```
{{t → 0}, {t → 0}}
```

<u>Adar infinituak</u>

```
to = -1;
Limit[x[t], t → to]
Limit[y[t], t → to]
```

$-\infty$

∞

```
to = -∞;
Limit[x[t], t → to]
Limit[y[t], t → to]
```

0

0

```
to = ∞;
Limit[x[t], t → to]
Limit[y[t], t → to]
```

0

0

★ **Koordenatu parametrikoko grafikoa**

```
ParametricPlot[{t/(1 + t^3), t^2/(1 + t^3)}, {t, -10, 10},
 ExclusionsStyle → Dashed, Exclusions → {1 + t^3 == 0},
 PlotStyle → {RGBColor[0.2, 0.8, 0.4], Thickness[0.008]}, PlotRange → {{-1, 1}, {-1, 1}}]
```



▼ Asintotadun funtzio parametrikoa

★ **Parametrizazioaren definizioa**

```
r[t_] = {x[t_], y[t_]} = {(t^2 + t)/(1 - t), (2 * t)/(1 - t)};

GraphicsArray[
 {Plot[x[t], {t, -3, 3}, PlotStyle → {RGBColor[0.7, 0.3, 0.2], Thickness[0.01]}],
  Plot[y[t], {t, -3, 3}, PlotStyle → {RGBColor[0.2, 0.3, 0.8], Thickness[0.01]}]}]
```



★ **Ebaki puntuen eta adar infinituen azterketa**

__Ebaki puntuak__

```
Solve[x[t] == 0, t]
```

{{t → −1}, {t → 0}}

```
Solve[y[t] == 0, t]
```

```
{{t → 0}}
```

Adar infinituak

```
to = 1;
Limit[x[t], t → to]
Limit[y[t], t → to]
```

```
-∞

-∞
```

```
to = -∞;
Limit[x[t], t → to]
Limit[y[t], t → to]
```

```
∞

-2
```

```
to = ∞;
Limit[x[t], t → to]
Limit[y[t], t → to]
```

```
-∞

-2
```

★ Koordenatu parametrikoko grafikoa

```
ParametricPlot[{(t^2 + t)/(1 - t), (2 * t)/(1 - t)}, {t, -10, 10}, ExclusionsStyle → Dashed,
  Exclusions → {t == 1}, PlotStyle → {RGBColor[0.2, 0.8, 0.4], Thickness[0.008]},
  PlotRange → {{-10, 8}, {-6, 4}}];
```

```
Manipulate[ParametricPlot[{(t^2 + t)/(1 - t), (2 * t)/(1 - t)}, {t, -10, a}, ExclusionsStyle → Dashed,
  Exclusions → {t == 1}, PlotStyle → {RGBColor[0.2, 0.8, 0.4], Thickness[0.008]},
  PlotRange → {{-10, 8}, {-6, 4}}], {a, -9.95, 10, 0.05}]
```



4.6. 3D-n parametrizatutako grafikoak

▼ ParametricPlot3D

Helikoidea

```
ParametricPlot3D[{Sin[u], Cos[u], u / 10}, {u, 0, 20},
 PlotStyle → Directive[Red, Thick], ColorFunction → "DarkRainbow"]
```